
Analysis of an Eddy Current Brake with FEMM

David Meeker

dmeeker@ieee.org

```
In[1]:= << c:\femm42\mathfemm\mathfemm.m
```

MathFEMM loaded at Sun 7 Apr 2013 11:59:18


■ Preliminary Definitions

```
In[2]:= mm = 0.001;
      Inch = 0.0254;
      MS = 10^6;
      MA = 10^6;
      m = 1;
       $\mu_o = 4 \cdot \text{Pi} \cdot 10^{-7}$ ;
      Hz = 2 * Pi;
```

■ Problem Parameters

A conductive plate moves within a double-sided array of steel-backed permanent magnets. This notebook computes the force of the brake vs. plate velocity.

One wavelength of the geometry, neglecting the end effects, looks like so:



```

In[9]:= BHmax = 40; (* Energy product of the magnet in MGOe *)
pp = 25 * mm; (* Pole pitch of the magnets *)
wm = 20 * mm; (* Magnet width *)
tm = 5 * mm; (* Magnet thickness *)
tb = 5 * mm; (* Back iron thickness *)
tp = 2.5 * mm; (* Conductive plate half-thickness *)
g = 1 * mm; (* Air gap length between magnet and plate *)
σ = 30; (* Plate conductivity in MS/m *)
h = 100 * mm; (* Depth of magnet array *)
ε = 20 * mm; (* Plate overhang on either side of the magnets *)

```

■ **Derived Quantities**

Mechanical frequency of the magnet array

```

In[19]:= β = π / pp;

```

A big problem with using a 2D solver to analyze this sort of machine is that 3D transverse edge effects can significantly influence the performance of the brake. Fringing near the transverse edge means that some of the PM flux doesn't link the plate, lessening the maximum force produced by the brake. The resistance of the edge of the plate, where the currents "turn around", makes the plate look more resistive and pushes out the speed at which the peak force occurs. Here, we will put in some kludgy corrections to account for these effects.

The total stack-up of unit permeability materials between plate centerline and iron is gtot:

$$\text{In[20]:= } \mathbf{gtot} = (\mathbf{tp} + \mathbf{g} + \mathbf{tm}) ;$$

Use this distance to get an approximate derating to account for fringing at the transverse edges. This is an approximate kludge that is probably no good for an array that isn't wide compared to gtot. Say that the width of a fringing region is hf:

$$\text{In[21]:= } \mathbf{hf} = \mathbf{gtot} / 2 ;$$

Shorten the active region by a bit, snipping off the fringing region of width hf on each side of the stator:

$$\text{In[22]:= } \mathbf{heff} = \mathbf{h} - 2 * \mathbf{hf} ;$$

Add the width of the fringing region to the width of the "end turn" region of the plate

$$\text{In[23]:= } \mathbf{eeff} = \mathbf{\epsilon} + \mathbf{hf} ;$$

Correct the conductivity to account for the transverse edges of the plate

$$\text{In[24]:= } \mathbf{\sigma eeff} = \mathbf{\sigma} / \left(1 + 2 / \left(\mathbf{\beta}^2 * \mathbf{heff} * \mathbf{eeff} \right) \right)$$

Out[24]= 28.3801

Remanence of the PMs in Tesla (see <http://www.femm.info/Archives/misc/BarMagnet.pdf>)

$$\text{In[25]:= } \mathbf{Br} = \text{Sqrt}[\mathbf{BHmax}] / 5 .$$


Out[25]= 1.26491

Coercivity of the PMs in A/m

$$\text{In[26]:= } \mathbf{Hc} = \mathbf{Br} / \mu_0$$

Out[26]= 1.00658×10^6

You can replace a permanent magnet by an equivalent current sheet that produces the same external fields. For this particular case, the equivalent current sheet representation looks like:



We want to replace the current sheet by a sinusoidally distributed band of current density. Most of the force in an eddy current brake is due to the fundamental, and if we just consider the fundamental, we can make a simulation in FEMM that includes "motion".

Find amplitude of the fundamental in a Fourier Series (http://en.wikipedia.org/wiki/Fourier_series) representation of the equivalent current density of the PMs in units of A/m².

There is a + current sheet of strength H_c located at $(pp-wm)/2$ and a - current sheet located at $(pp+wm)/2$. The result is in units of A/m²

```
In[27]:= Jmag = (Hc * Cos[β * (pp - wm) / 2] - Hc * Cos[β * (pp + wm) / 2]) / Integrate[Cos[β x]^2, {x, 0, pp}]
```

```
Out[27]= 1.53171 × 108
```

The analysis approach is to analyze the brake from the point of view of an observer attached to the plate. From the point of view of this observer, the magnet's sinusoidally distributed current bands look like simple AC currents. We can analyze the problem in FEMM by defining a distribution of currents in the region where the permanent magnets live that looks like $Jmag \cdot \exp(i \cdot \beta \cdot x)$ where x indexes the position down the array. The apparent frequency is $\omega = \beta \cdot v$, where v is the plate's velocity.

■ Draw Domain

This section uses the MathFEMM interface between Mathematica and FEMM to programmatically draw the brake geometry.

```
In[28]:= OpenFEMM[]
```

```
In[29]:= NewDocument[0]
```

Define Materials

```
In[30]:= MIGetMaterial["Air"];
```

```
In[31]:= MIAddMaterial["Steel", 5000, 5000];
```

```
In[32]:= MIAddMaterial["Plate", 1, 1, 0, 0,oeff, 0, 0, 1, 0, 0, 0, 1, 1];
```

Draw Plate

```
In[33]:= MIDrawRectangle[{0, 0}, {2 pp, tp}];
MIAddBlockLabel[{pp, tp / 2}];
MISelectLabel[{pp, tp / 2}];
MISetBlockProp["Plate", 1, 0, "<None>", 0, 1, 1];
MIClearSelected[];
```

Draw Air Gap

```
In[38]:= MIDrawRectangle[{0, tp}, {2 * pp, tp + g}];
MIAddBlockLabel[{pp, tp + g / 2}];
MISelectLabel[{pp, tp + g / 2}];
MISetBlockProp["Air", 1, 0, "<None>", 0, 1, 1];
MIClearSelected[];
```

Draw Magnets

```
In[43]:= n = 12;
For[k = 1, k ≤ n, k++,
  xl = 2 * (k - 1) * pp / n;
  xr = 2 * k * pp / n;
  xm = (xl + xr) / 2;
  MIAddMaterial["J" <> ToString[k], 1,
    1, 0, (Jmag / MA) * Exp[I * β * xm], 0, 0, 0, 1, 0, 0, 0, 1, 1];
  MIDrawRectangle[{xl, tp + g + tm}, {xr, tp + g}];
  MIAddBlockLabel[{xm, tp + g + tm / 2}];
  MISelectLabel[{xm, tp + g + tm / 2}];
  MISetBlockProp["J" <> ToString[k], 1, 0, "<None>", 0, 0, 1];
  MIClearSelected[];
];
```

Draw Backiron

```
In[45]:= MIDrawRectangle[{0, tp + g + tm}, {2 * pp, tp + g + tm + tb}];
MIAddBlockLabel[{pp, tp + g + tm + tb / 2}];
MISelectLabel[{pp, tp + g + tm + tb / 2}];
MISetBlockProp["Steel", 1, 0, "<None>", 0, 0, 1];
MIClearSelected[];
```

Add Boundary Conditions

```
In[50]:= MIAddBoundProp["A=0", 0, 0, 0, 0, 0, 0, 0, 0, 0];
MIAddBoundProp["pbc1", 0, 0, 0, 0, 0, 0, 0, 0, 4];
MIAddBoundProp["pbc2", 0, 0, 0, 0, 0, 0, 0, 0, 4];
MIAddBoundProp["pbc3", 0, 0, 0, 0, 0, 0, 0, 0, 4];
MIAddBoundProp["pbc4", 0, 0, 0, 0, 0, 0, 0, 0, 4];
```

```
In[55]:= MISelectSegment[{pp, tp + g + tm + tb}];
MISetSegmentProp["A=0", 0, 1, 0, 0];
MIClearSelected[];
```

```
In[58]:= MISelectSegment[{0, tp + g + tm + tb / 2}];
MISelectSegment[{2 * pp, tp + g + tm + tb / 2}];
MISetSegmentProp["pbc1", 0, 1, 0, 0];
MIClearSelected[];
```

```

In[62]:= MISelectSegment[{0, tp + g + tm / 2}];
MISelectSegment[{2 * pp, tp + g + tm / 2}];
MISetSegmentProp["pbc2", 0, 1, 0, 0];
MIClearSelected[];

In[66]:= MISelectSegment[{0, tp + g / 2}];
MISelectSegment[{2 * pp, tp + g / 2}];
MISetSegmentProp["pbc3", 0, 1, 0, 0];
MIClearSelected[];

In[70]:= MISelectSegment[{0, tp / 2}];
MISelectSegment[{2 * pp, tp / 2}];
MISetSegmentProp["pbc4", 0, 1, 0, 0];
MIClearSelected[];

In[74]:= MIZoomNatural[]

```

■ Calculate Force vs Speed

Expected speed at which maximum braking force occurs in m/s, based on 1D linear induction motor theory:

```

In[75]:=  $\omega c = (\beta^2 * g_{tot}) / ((\sigma_{eff} * MA / m) * \mu_o * tp)$ 

```

```

Out[75]= 1505.48

```

```

In[76]:=  $v_{opt1d} = \omega c / \beta$ 

```

```

Out[76]= 11.9802

```

We can use the quick analytical estimate of v_{opt} to select the range over which we evaluate the performance of the brake.

```


In[77]:= MISaveAs[NotebookDirectory[] <> "temp.fem"];

In[78]:= data = {};
For[v = 0, v < 2 * vopt1d, v++,
  MIProbDef[ $\beta * v$  / Hz, "meters", "planar", 10(-8), heff, 30];
  MIAnalyze[];
  MILoadSolution[];
  MOGroupSelectBlock[1];
  f = MOBlockIntegral[11];
  data = Append[data, {v, f}]
];
v = .

In[81]:= CloseFEMM[]

```

```
In[82]:= ListPlot[data, Joined → True, ImageSize → 500, Frame → True,
  GridLines → {Table[k, {k, 0, 25, 5}], Table[-k, {k, 0, 600, 100}]},
  FrameLabel → {"Speed, m/s", "Force, N"}, PlotStyle → {Directive[Thick]},
  BaseStyle → {FontFamily → "ComicSansMS", FontSize → 14}]
```



```
In[83]:= opt = FindMinimum[Interpolation[data][vopt], {vopt, 5, 15}]
```

```
Out[83]= {-538.157, {vopt → 10.2748}}
```


```
In[84]:= Fmax = 2 opt[[1]];
  vopt = vopt /. opt[[2]];
```

Theoretically, the braking force out to have the analytical form :

```
In[86]:= Fbrake = Fmax * (v / vopt) / (1 + (v / vopt) ^ 2)
```

```
Out[86]= -  $\frac{104.752 v}{1 + 0.00947217 v^2}$ 
```

```
In[87]:= Show[ListPlot[data, PlotRange -> All],
  Plot[Fbrake, {v, 0, 2 vopt1d}, PlotRange -> All], ImageSize -> 500, Frame -> True,
  GridLines -> {Table[k, {k, 0, 25, 5}], Table[-k, {k, 0, 600, 100}]},
  FrameLabel -> {"Speed, m/s", "Force, N"},
  BaseStyle -> {FontFamily -> "ComicSansMS", FontSize -> 14}]
```



■ Conclusions

A basic analysis of a "long stator" eddy current brake has been presented in this notebook. End effects have been neglected. Long stator end effects could be modeled by simply modeling the entire length of the plate in the present style of analysis, adding a few extra wavelengths of stator on either side of the plate. End effects make a force ripple at a frequency of $2\beta v$, but for a reasonably long brake, the amplitude of the ripple is relatively small and the change to the DC force is negligible. See, for example,

http://esrdc.mit.edu/library/ESRDC_library/Kirtley_The_End_Effect_in_Short_S_1.pdf

<http://www.dtic.mil/dtic/tr/fulltext/u2/a447307.pdf>

<http://ieeexplore.ieee.org/iel5/27/5680667/05613198.pdf>

However, for "short stator" machines (e.g. an array of magnets riding down an aluminum fin), the end effects are a bit more significant. A solver that includes motion is really needed to analyze the end effects on a short stator machine. See, for example:

<http://www.jeet.or.kr/ltkpsweb/pub/pubfpfile.aspx?ppseq=31>

<http://onlinelibrary.wiley.com/doi/10.1002/eej.4390980407/abstract>